
アプリケーション開発

クリスマスツリー


【目次】

クリスマスツリーを作る.....	3
透明のフォームを作ります.....	3
電球の玉を光らせます.....	6
ツリーを動かせるようにします.....	8

クリスマスツリーを作る

クリスマスも近くなってきました。デスクトップにクリスマスツリーを表示してみましょう。

透明のフォームを作ります



サイズは、70,160。TransparentKey を Black にします。その中に Timer を2つ貼り付けて、Timer1 を Interval=100,Timer2 を Interval=1000 にします。その中に pictureBox を入れて、同じサイズにします。

そして、Form_Load に以下のコーディングをして、まずツリーの緑と鉢の茶色を書きます。

```
PictureBox1.Image = New Bitmap(70, 160)
' imgのGraphicsオブジェクトを取得
g = Graphics.FromImage(PictureBox1.Image)

' 緑に塗りつぶす
g.FillRectangle(Brushes.Green, 0, 0, 70, 140)

' 鉢を茶色に塗りつぶす
g.FillRectangle(Brushes.Brown, 20, 140, 50, 160)
```

public の宣言を先に書いておきましょう

```
Public g As Graphics
Public rnd As New Random()
Public pntX(30) As Integer
Public pntY(30) As Integer
Public pntCir(30) As Integer
Public pntW As Integer = 0
Public pntXr(40) As Integer
Public pntYr(40) As Integer
Public pntCr(40) As SolidBrush
Public pntR As Integer = 0

' マウスポジションを保持
Private mousePoint As Point
```

続いて、フォームの形をツリーの形に切り抜きます。Form_Load につづいてコーディングします。

```
' フォームのサイズを適当に変更
' 多角形の頂点の位置を設定
Dim points() As Point = {
    New Point(35, 0),
    New Point(0, 140),
    New Point(20, 140),
    New Point(20, 160),
    New Point(50, 160),
    New Point(50, 140),
    New Point(70, 140)
}
Dim types() As Byte = {
    Drawing.Drawing2D.PathPointType.Line,
    Drawing.Drawing2D.PathPointType.Line,
    Drawing.Drawing2D.PathPointType.Line,
    Drawing.Drawing2D.PathPointType.Line,
    Drawing.Drawing2D.PathPointType.Line,
    Drawing.Drawing2D.PathPointType.Line,
    Drawing.Drawing2D.PathPointType.Line,
    Drawing.Drawing2D.PathPointType.Line
}
' GraphicsPathの作成
Dim path As New Drawing2D.GraphicsPath(points, types)
' 形を変更
Me.Region = New Region(path)
```

電球の玉を光らせます

白い玉を 28 個。いろいろな色の玉を 40 個書き込みます。

```
Me.Left = 5
Me.Top = 570

Dim j As Integer = 0
For i = 0 To 140 Step 5
    pntCir(j) = rnd.Next(8) + 1
    pntY(j) = i
    pntX(j) = rnd.Next(i / 2) + ((70 / 2) - ((35 / 140) * i))
    g.FillEllipse(Brushes.White, pntX(j), pntY(j), pntCir(j), pntCir(j))
    j += 1
Next i

For i = 0 To 39
    '色の作成
    Dim col As New SolidBrush(Color.FromArgb(rnd.Next(256), rnd.Next(256),
rnd.Next(256)))
    pntCr(i) = col
    '1つのピクセルの色を変える
    pntYr(i) = rnd.Next(140)
    pntXr(i) = rnd.Next(pntYr(i) / 2) + ((70 / 2) - ((35 / 140) * pntYr(i)))
    g.FillEllipse(col, pntXr(i), pntYr(i), 3, 3)
Next i

PictureBox1.Invalidate()

Timer1.Start()
Timer2.Start()
```

Timer1 は、白い玉が順々に点滅するものです

```
Private Sub Timer1_Tick(sender As Object, e As EventArgs) Handles Timer1.Tick
    g.FillEllipse(Brushes.White, pntX(pntW), pntY(pntW), pntCir(pntW), pntCir(pntW))
    pntW += 1
    If pntW > 28 Then pntW = 0
    g.FillEllipse(Brushes.Green, pntX(pntW), pntY(pntW), pntCir(pntW), pntCir(pntW))

    PictureBox1.Invalidate()
End Sub
```

Timer2 は球が一斉に点滅します。

```
Private Sub Timer2_Tick(sender As Object, e As EventArgs) Handles Timer2.Tick
    If pntR = 0 Then
        For i = 0 To 39
            g.FillEllipse(pntCr(i), pntXr(i), pntYr(i), 3, 3)
        Next i
        pntR = 1
    Else
        For i = 0 To 39
            g.FillEllipse(Brushes.Green, pntXr(i), pntYr(i), 3, 3)
        Next i
        pntR = 0
    End If
    PictureBox1.Invalidate()
End Sub
```

ツリーを動かせるようにします

そして、ツリー自体を動かせるようにします。

Form1のMouseDownとMouseMoveイベントを作成してから中身をコーディングします。

```
Private Sub Form1_MouseDown(ByVal sender As Object, _
    ByVal e As System.Windows.Forms.MouseEventArgs) _
    Handles MyBase.MouseDown, PictureBox1.MouseDown
    If (e.Button And MouseButtons.Left) = MouseButtons.Left Then
        '位置を記憶する
        mousePoint = New Point(-e.X, -e.Y)
    End If
End Sub

'ピクチャーのイベントをフォームにすりかえる
Private Sub Form1_MouseMove(ByVal sender As Object, _
    ByVal e As System.Windows.Forms.MouseEventArgs) _
    Handles MyBase.MouseMove, PictureBox1.MouseMove
    If (e.Button And MouseButtons.Left) = MouseButtons.Left Then
        Me.SetDesktopLocation( _
            Me.Left + mousePoint.X + e.X, _
            Me.Top + mousePoint.Y + e.Y)
    End If
End Sub
```

終了は Form1_DoubleClick イベントで End します

```
'終了
Private Sub Form1_DoubleClick(sender As Object, e As EventArgs) Handles
Me.DoubleClick, PictureBox1.DoubleClick
    End
End Sub
```

あとは自分なりにタイミングを変えたり、数を変えたりしてみましょう。